

# 「EQUATRAN-M」技術計算用連立方程式解法言語

(3)

## EQUATRAN-M の化工計算への応用

須藤 精一\*  
佐渡 友秀\*\*  
宮原 景中\*\*\*

前回までで EQUATRAN-M の文法と使用法を例題に沿って説明したが、今回は各種化工計算への応用について解説する。なお、熱交換器の設計と解析、プロセスフローシミュレーションについては回を改めて詳しく述べることにし、ここでは日頃よく使われる基礎的な化工計算のいくつかを選んで、EQUATRAN-M を使えば手軽に計算が行えることを示す。

### 1. 平衡計算

化工計算の基礎として重要なイオン平衡 (pH 値) 及び化学平衡の計算例を紹介する。

#### 1.1 pH の計算

pH の計算はよく必要になるが、多塩基酸などを含む溶液の場合にはいささか面倒である。イオン平衡の計算については文献<sup>1)</sup>に極めて有用な方法が発表されているが、EQUATRAN-M によると簡単に解けることを示そう。

#### <例題 3.1: pH の計算>

0.01 mol/l の炭酸水溶液 1l, 0.01 mol/l のアンモニア水溶液 1l, 0.01 mol/l のリン酸水溶液 3l の混合液の pH はいくらか<sup>1)</sup>。ただし、解離定数は次のとおり。

\* Seiichi Sudo 三井東圧化学(株)システム部

\*\* Hideo Sadotomo 三井東圧化学(株)システム部 主席部員

\*\*\* Koreatsu Miyahara 三井東圧化学(株)システム部次長

$$\begin{aligned} \frac{[\text{H}^+][\text{HCO}_3^-]}{[\text{H}_2\text{CO}_3]} &= 10^{-6.3}, \quad \frac{[\text{H}^+][\text{CO}_3^{2-}]}{[\text{HCO}_3^-]} \\ &= 10^{-10.1} \\ \frac{[\text{NH}_4^+][\text{OH}^-]}{[\text{NH}_4\text{OH}]} &= 10^{-9.25} \\ \frac{[\text{H}^+][\text{H}_2\text{PO}_4^-]}{[\text{H}_3\text{PO}_4]} &= 10^{-2}, \quad \frac{[\text{H}^+][\text{HPO}_4^{2-}]}{[\text{H}_2\text{PO}_4^-]} \\ &= 10^{-6.9}, \quad \frac{[\text{H}^+][\text{PO}_4^{3-}]}{[\text{HPO}_4^{2-}]} = 10^{-11.7} \end{aligned}$$

解) リスト及び計算結果を表 3-1 に示す。与えられた溶液の体積、濃度及び混合液の体積 (V) を 4~6 行に書く。これらはコメントから自明であろう。次に、記法上、上つき + - 等は許されないのを省略した平衡の式を 9~16 行に示す。これに水のイオン積 (19行) と元素 C, N, P についての収支式 (22~24行) が続いている。元素収支は濃度を mol/l 単位とするため (Va l/V) などで補正してある。若干の説明を要するとすればイオン収支であろう。酸由来の H<sup>+</sup> のうち α だけが塩基由来の OH<sup>-</sup> と結合すると仮定すると、

$$\begin{aligned} &[\text{HCO}_3^-] + 2[\text{CO}_3^{2-}] + [\text{H}_2\text{PO}_4^-] \\ &+ 2[\text{HPO}_4^{2-}] + 3[\text{PO}_4^{3-}] \\ &= \text{H} \text{ (27行目の)} + \alpha \\ &[\text{NH}_4^+] = \text{OH} \text{ (27行目の)} + \alpha \quad (3-1) \end{aligned}$$

となることは明らかであろう。この両式から α を消去すると 27 行の式が得られる。

29 行目までで式は完結しているが、pH は 0~14 の範囲であることを与えないと負の領域まで探

表3-1 例題3.1の解

```

1: /* pH の計算 */
2:
3: /* H2CO3 NH3 H3PO4 */
4: Va1=1 ; Vb1=1 ; Va2=3 /*体積 [1] */
5: Ca1=0.01; Cb1=0.01; Ca2=0.01/*濃度 [mol/l]*/
6: Va1 + Vb1 + Va2 =V /*溶液体積 [1]*/
7:
8: /* 解離平衡 */
9: H*HCO3/H2CO3 = 10^(-6.3) /* H2CO3 */
10: H*CO3/HCO3 = 10^(-10.1)
11:
12: NH4*OH/NH4OH = 10^(-9.25) /* NH3 */
13:
14: H*H2PO4/H3PO4 =10^(-2)
15: H*HPO4 /H2PO4 =10^(-6.9) /* H3PO4 */
16: H*PO4 /HPO4 =10^(-11.7)
17:
18: /* 水のイオン積 */
19: H*OH = 1.0E-14
20:
21: /* 元素収支 */
22: H2CO3+HCO3+CO3 = Ca1*(Va1/V) /* C */
23: NH4OH+NH4 = Cb1*(Vb1/V) /* N */
24: H3PO4+H2PO4+HPO4+PO4 = Ca2*(Va2/V)/* P */
25:
26: /* イオン収支 */
27: e: HCO3+2*CO3+H2PO4+2*HPO4+3*PO4+OH = NH4+H
28:
29: pH = -LOG10(H)
30:
31: reset pH # 2 [ 0,14 ] BY e
    
```

[ 計算結果 ]

```

Va1 = 1
Vb1 = 1
Va2 = 3
Ca1 = 0.01
Cb1 = 0.01
Ca2 = 0.01
V = 5
H2CO3 = 0.002
HCO3 = 3.673841E-007
H = 0.002728
CO3 = 1.069770E-014
NH4OH = 1.295323E-005
OH = 3.665811E-012
NH4 = 0.001937
H3PO4 = 0.001286
H2PO4 = 0.004714
HPO4 = 2.175447E-007
PO4 = 1.591178E-016
pH = 2.56417
    
```

索してしまう。このため31行を追加する。表3-1のリストを見れば非常に簡単に解が得られることが判るであろう。なお文献<sup>2)</sup>による pH の計算値は2.568±0.005である。

## 1.2 化学平衡

実際の化学プロセスにおける反応では、平衡計算により、反応条件、達成可能な最大収率等重要な情報が得られる。現実には複数の反応が並行して起こることが多く、そのような場合に平衡組成を求めるのは大変労力のいることである。

### <例題 3.2: 化学平衡計算>

鉄製反応器内の炭素塊の層を水蒸気が通過する(反応条件: 1 atm, 600°C)。生成物の反応平衡組成を求めよ<sup>2)</sup>。このとき、起こり得る反応として多数考えられるが、次の3つの独立な

表3-2 例題3.2の解

```



1: /* 化学平衡 反応平衡組成の計算 */
2: /* O.A.Hougen,K.M.Watson,R.A.Ragatz :
3: Chemical Process Principles P.1053 */
4: H2O = 1 - x - z
5: H2 = x - 2*y + z
6: CO2 = x
7: CO = - x + z
8: CH4 = y
9: H2O + H2 + CO2 + CO + CH4 = total
10: eq1 : K1 = CO*H2 / (H2O*total)
11: eq2 : K2 = CO2*H2 / (CO*H2O)
12: eq3 : K3 = CH4*total / (H2^2)
13: K1 = 0.269 ; K2 = 2.21 ; K3 = 0.470
14: RESET x#0.1 [0,1] BY eq1
15: RESET y#0.2 [0,1] BY eq2
16: RESET z#0.5 [0,1] BY eq3
17: OUTPUT K1,K2,K3,x,y,z,H2O,H2,CO2,CO,CH4
    
```

[ 計算結果 ]

```

K1 = 0.269
K2 = 2.21
K3 = 0.47
x = 0.252215
y = 0.093771
z = 0.456807
H2O = 0.290978
H2 = 0.521481
CO2 = 0.252215
CO = 0.204591
CH4 = 0.093771
    
```

反応により量論関係が完全に記述できる。



各反応の平衡定数は、

$$K_1 = \frac{[\text{CO}][\text{H}_2]}{[\text{H}_2\text{O}][\text{total}]} = 0.269 \quad (3-2)$$

$$K_2 = \frac{[\text{CO}_2][\text{H}_2]}{[\text{CO}][\text{H}_2\text{O}]} = 2.21 \quad (3-3)$$

$$K_3 = \frac{[\text{CH}_4][\text{total}]}{[\text{H}_2]^2} = 0.470 \quad (3-4)$$

ここで、反応前の水蒸気量を 1 mol とし、

$x$  = 反応[1]で消失した水蒸気量 [mol]

$y$  = 反応[3]で生成したメタン量 [mol]

$z$  = 反応[2]で消失した水蒸気量 [mol]

と表わせば、平衡時の各成分量は、

$$[\text{H}_2\text{O}] = 1 - x - z \quad (3-5)$$

$$[\text{H}_2] = x - 2y + z \quad (3-6)$$

$$[\text{CO}_2] = x \quad (3-7)$$

$$[\text{CO}] = z - x \quad (3-8)$$

$$[\text{CH}_4] = y \quad (3-9)$$

$$[\text{total}] = 1 - y + z \quad (3-10)$$

となる。[total] はガス成分の全モル数である。

解) (3-2)~(3-4)式の3つの非線形連立方程式を

解く。リスト、計算結果を表3-2に示す。リスト中の式の意味はほとんど明らかであろう。独立変数  $x, y, z$  は0と1の間の値をとることがわかっているが、初期値の与え方が大事である。物理的考察により、少なくとも(3-5)~(3-9)式の各成分が正の条件を満たすように選ぶべきである。(14~16行)なお、本例題では計算結果に示した解以外に、 $[H_2]$ 、 $[CO]$  が負となる物理的に無意味な解が存在し、初期値によってはその解へ収束するので注意を要する。

## 2. 状態方程式

状態方程式は単に PVT 関係の推算に限らず、相平衡の計算、エンタルピー等種々の物性計算に利用される。高压での物質の PVT 関係を表わすために古くから状態方程式が研究され、Redlich-Kwong 式、BWR 式、Peng-Robinson 式など多数の式が提案されているが、少ないパラメータで近似度の高い Soave-Redlich-Kwong 状態方程式(以下 SRK 式と略記)がよく用いられる。

### <例題 3.3 : SRK 状態方程式>

SRK 式<sup>3)</sup>により対臨界温度、圧力:  $T_r, P_r$  を与えて圧縮係数  $z(=PV/RT)$  を計算する。SRK 式は、

$$z = \frac{V}{V-b} - \frac{\Omega_a}{\Omega_b} \frac{b}{V+b} F \quad (3-11)$$

ただし、

$$F = \frac{1}{T_r} [1 + (0.480 + 1.574 \omega - 0.176 \omega^2)(1 - T_r^{0.5})]^2$$

$$b = \Omega_b RT_c / P_c$$

で、 $\Omega_a, \Omega_b$  は定数である。

解) リスト及び計算結果を表3-3に示す。

$$A = (\Omega_a / \Omega_b) B F = \Omega_a F P_r / T_r$$

$$B = b P / RT = \Omega_b P_r / T_r$$

とおくと(3-11)式は  $z$  に関する3次式

$$z^3 - z^2 + (A - B - B^2)z - AB = 0$$

となり、係数( $T_r, P_r$  の関数)の値により解が複数存在しうる。EQUATRAN-Mでは同時にすべての解を求めることはできないが、独立変数の初期値、変域を適当に選ぶことによりすべての解を求めることが可能である。例えば本例題(プロパン)では、 $T_r=0.9, P_r=0.5$  のとき3根(小さ

表3-3 例題3.3の解

```

1: /* SOAVE-REDLICH-KWONG 状態方程式 */
2: P*V = z*R*T
3: eq: z = V/(V-b) - (OMGa/OMGb)*b/(V+b)*F
4: b = OMBb*R*Tc/Pc
5: R = 0.08205
6: OMGa = 0.4274802327
7: OMBb = 0.086640350
8: F = (1+(0.480+1.574*omg-0.176*omg^2) ..
9:      *(1-SQRT(Tr)))^2/Tr
10: Tr = T/Tc ; Pr = P/Pc
11: RESET z #0.5[0,1] BY eq
12: /* for Propane : The Properties of Gases
13:    and Liquids 3rd Ed. p.639 (1977) */
14: Tc = 369.8 ; Pc = 41.9 ; omg = 0.152
15: INPUT Tr,Pr
16: OUTPUT Tr,Pr,z,V,F

```

[ 入力値(=) または 初期値(#) ]

Tr = 0.9  
Pr = 0.5  
z = # 0.5

[ 計算結果 ]

Tr = 0.9  
Pr = 0.5  
z = 0.691341  
V = 0.901148  
F = 1.194165

い順に0.090, 0.218, 0.691) 存在するが、初期値を変える(例えば0.1, 0.2, 0.5) ことにより3根とも求まる。SRK 式では、2相共存領域ではこのように3根存在するが、その最大の根は飽和気体、最小のものは飽和液体の圧縮係数に対応する。真中の解は trivial な解である。なお、3次方程式の解法としてよく知られているものに Cardan の方法<sup>4),5)</sup>がある。上記の解はこの方法による3つの解に一致している。

ここでは、純物質の状態の計算法を例示したが、適当な混合則と2成分パラメータ  $k_{ij}$  を導入すれば<sup>17)</sup>、2成分以上の系にも適用できることは言うまでもない。

## 3. 気液平衡式のパラメータ決定

非理想溶液の液相活量係数を表現するのに、van Laar 式、Wilson 式、NRTL 式その他の式が数多く発表されているが、中でも Wilson 式はその適用範囲の広さから多用されている。しかし、実測データからそのパラメータを求めることは意外に面倒である。ここでは無限希釈溶液の活量係数からのパラメータ決定法について述べる。

### <例題 3.4 : Wilson パラメータの計算>

無限希釈における2成分系 Wilson 式、

$$\ln \gamma_1^0 = -\ln A_{12} + (1 - A_{21}) \quad (3-12)$$

$$\ln \gamma_2^0 = -\ln A_{21} + (1 - A_{12}) \quad (3-13)$$

において、 $\gamma_1^0, \gamma_2^0$  を与えてパラメータ  $A_{12}, A_{21}$  を求めよ。

解) (3-12), (3-13) 式は一般的な 2 成分系 Wilson 式に比べれば簡単な式であるが、非線形であり、また  $\gamma_1^0, \gamma_2^0$  の値によっては解が複数存在する。この問題に対しては宮原らのノモグラフ<sup>6)</sup>を使用するのが便利である(図 3-1 参照)が、さらに精度を要する場合には (3-12), (3-13) 式を数値的に解けばよい。そのためには  $A_{12}$  または  $A_{21}$  の適当な初期値、変域を指定する必要があるが、上述のノモグラフはそれらの非常に良い推定値となる。EQUATRAN-M による計算例を表 3-4 に示す。ここで  $\gamma_1^0 = 0.4, \gamma_2^0 = 0.56$  とする。このケースでは図 3-1 に示されているように解が 3 組存在するが、例題 3.3 と同様に初期値と変域を変えることにより 3 組とも求められる。

なお、EQUATRAN-M では、必要であれば繰り返し計算、最適化計算の途中経過を画面または




図 3-1 Wilson パラメータの算出<sup>6)</sup>

プリンターへ打ち出すことができるので未収束の場合の検討に便利である。表 3-4 には繰り返し計算の場合の出力例を示した。出力の内容は左から独立変数値、収束チェックを行なう式の左辺と右辺の値及びその誤差である。

多数のデータから最小二乗法による Wilson パラメータ決定法への EQUATRAN-M の利用については文献<sup>7)</sup>に例示されている。

#### 4. フラッシュ計算

フラッシュ分離器は気液平衡関係を利用した最も基本的な成分分離装置であり、相対揮発度が大きい場合にはこれだけで目的の分離が達成されるので化学プロセスではよく見かけられる。

##### <例題 3.5: 平衡フラッシュ蒸留>

プロパン 10 mol %, ブタン 59 mol %, ペンタン 31 mol % の混合物 100 mol を 10 atm, 93.5°C において平衡状態に保つとき(すなわち平衡蒸留すれば)、液と蒸気の量および組成はどうなるか<sup>8)</sup>。

解) この条件でプロパン、ブタン、ペンタンの平衡比はそれぞれ 2.80, 1.30, 0.55 である。

$$L + V = F \quad (3-14)$$

$$Lx_i + Vy_i = f_i \quad (i=1\sim 3) \quad (3-15)$$

$$y_i = K_i x_i \quad (i=1\sim 3) \quad (3-16)$$

ここで、 $F, f_i$  は全フィード量及び第  $i$  成分フィード量 [mol],  $L, V$  は液及び蒸気量 [mol],  $x_i, y_i$  は液相及び気相組成,  $K_i$  は平衡比である。また、組成に関する条件から

$$\sum x_i = 1 \quad \text{または} \quad \sum y_i = 1 \quad (3-17)$$

これを解くには上記 4 式をそのまま EQUATRA

表 3-4 例題 3.4 の解

```

1: /* Wilson パラメータの計算 */
2: /* 無限希釈溶液の活量係数から Wilson パラメータを求める */
3: J. Chem. Eng. Japan, Vol. 3, No. 2, p. 157 (1970) 参照
4: g10, g20 : 無限希釈溶液の活量係数
5: A12, A21 : Wilson パラメータ
6:
7: eq : loge(g10) = - loge(A12) + (1 - A21)
8:      loge(g20) = - loge(A21) + (1 - A12)
9:      g10 = 0.4
10:      g20 = 0.56
11: RESET A21 # 2.0 [1,3] BY eq
[ 途中経過 ]
<R 0> A21      = 2          -0.91629      =? -0.87972      :-0.03657
<R 1> A21      = 2.02       -0.91629      =? -0.88843      :-0.02786
<R 2> A21      = 2.08393    -0.91629      =? -0.91618      :-0.00011
[ 計算結果 ]
A21 = 2.083933
A12 = 0.845561
g10 = 0.4
g20 = 0.56

```

表 3-5 例題 3.5 の解

```

1: /* フラッシュ計算 */
2: /* 出典: 化学工学便覧 改訂四版 p.597(1978) */
3: (1)プロパン-(2)ブタン-(3)ペンタン
4: f: フィード量 [mol]
5: x: 液相組成
6: y: 気相組成
7: K: 平衡定数
8: VAR f(3),x(3),y(3),K(3)
9: L+V = SUM(f)
10: f = ( 10, 59, 31)
11: K = ( 2.8, 1.3, 0.55)
12: eq: SUM(K*f/(V*K+L))=1
13: RESET L # 50[0,100] BY eq UNTIL 0.01%
14: y = K*x
15: L*x + V*y = f
16: OUTPUT f,L,V,x,y
    
```

[ 計算結果 ]

f =			
1) 10	2) 59	3) 31	
L =	20.619797		
V =	79.380203		
x =			
1) 0.041172	2) 0.476521	3) 0.482273	
y =			
1) 0.115281	2) 0.619477	3) 0.26525	

N-M の記法で書けばよい. この方がいかにも EQUATRAN-M 的であるが, ここでは数式モデルの選定の重要性を示すために, 通常の計算法に従う. (3-15), (3-16) 式より  $x_i$  または  $y_i$  を消去し, (3-17) 式に代入することにより,

$$\sum \frac{K_i f_i}{K_i V + L} = 1 \quad (3-18)$$

または,

$$\sum \frac{f_i}{K_i V + L} = 1 \quad (3-19)$$

と変形し, いずれかの式で  $L$  または  $V$  を独立変数として収束計算を行う. あるいは (3-18), (3-19) 式を組み合わせた,

$$\sum \left( \frac{K_i f_i}{K_i V + L} - \frac{f_i}{K_i V + L} \right) = 0 \quad (3-20)$$

を使用してもよい. (3-18) 式を使用した場合のリスト及び計算結果を表 3-5 に示す. (3-19), (3-20) 式を使用する場合には 12 行の式をそれぞれ置き換えればよい.

このように (3-18)~(3-20) 式のいずれかの式を用いればよいが, 注意を要するのは用いる式により収束状況がかなり異なることである. これらの収束状況を図 3-2 に示すが, (3-18), (3-19) 式によった場合では真の解以外にそれぞれ  $L=0$ ,  $L=100$  という別の解(にせの解)が存在し, 初期値によってはその解へ収束してしまう. (3-20) 式を用いればそのようなことがなく, 他の 2 つに比べ収束性もよいことがわかる. これはちょっとした式の変形が収束性の向上をもたらす好例である. なお Lockhart らの方法<sup>9)</sup>による収束状況も図 3-2 中に示しておいた.

本例題は回分操作であるが, 連続系でも計算法に違いはない.

### 5. 簡易蒸留計算

連続蒸留塔の詳細設計に先立ち概略を把握するために, あるいは運転の合理化を検討する上で最小理論段・最小還流比は重要な目安となる. 厳密には精密蒸留計算を必要とするが, 概略値を知る上で簡易蒸留計算は大変有用である.

#### <例題 3.6: 簡易蒸留計算>

$n$ -ヘキサン(1)とトルエン(2)の混合物 (組成  $z_f$ ) を蒸留塔で下表のような留出組成 ( $x_d$ ) と缶出組成 ( $x_w$ ) に分離したい. いま, 塔は 760




図 3-2 用いる式による収束性の違い

mmHg で運転し, 原料は  $t_f=94.1^\circ\text{C}$  で平衡な気液混相 (組成  $y_f, x_f$ , 液の割合  $q$ ) で供給される. 最小還流比  $R_m$  に対する実還流比  $R$  の割合を  $R/R_m=1.2$  とするとき所要理論段数  $N$  と還流比を求めよ. なお, 蒸気圧は Antoine 式で表わし, 本系を理想溶液として扱う.

	$z_f$	$x_d$	$x_w$
$n$ -ヘキサン	0.35	0.95	0.10
トルエン	0.65	0.05	0.90

解) 第 1 成分に対する相対揮発度を  $\alpha_{1i}(i=1,2)$  とすると, 最小理論段  $N_m$  は Fenske の式

$$S_m = N_m + 1 = \frac{\log \left\{ \left( \frac{x_{d1}}{x_{d2}} \right) \left( \frac{x_{w2}}{x_{w1}} \right) \right\}}{\log \alpha^{av}} \quad (3-21)$$

で与えられる。ここで、 $S_m$  は最小ステップ数で、 $\alpha^{av}$  は塔頂と塔底における  $\alpha_{12}$  の幾何平均である。

$$\alpha^{av} = \sqrt{\alpha_{12}^{TOP} \alpha_{12}^{BOTTOM}} \quad (3-22)$$

また、最小還流比  $R_m$  は Underwood の方法により

$$\sum_{i=1}^2 \frac{Z_{fi}}{k - \alpha_{1i}} = \frac{1 - q}{k} \quad (3-23)$$

で  $k$  を解き ( $\alpha_{11} < k < \alpha_{12}$ )

$$R_m = k \left( \sum_{i=1}^2 \frac{x_{di}}{k - \alpha_{1i}} \right) - 1 \quad (3-24)$$

に代入して求められる。なお、ここでの  $\alpha_{1i}$  は文献<sup>16)</sup>に従がい原料供給段のそれを用いる。さらに、実ステップ数  $S (= N + 1)$  と還流比の関係は Gilliland の相関 (平田の式)

$$\log \frac{S - S_m}{S + 1} = -0.9 \frac{R - R_m}{R + 1} - 0.17 \quad (3-25)$$


でよく表わされることが知られている<sup>16)</sup>。

リストと計算結果を表 3-6 に示す。8 行から 12 行で原料の気液平衡関係をフラッシュ計算にて求めており (例題 3.5 で (3-15) 式両辺を  $F$  で割った式が 9 行目の式である)、15~18 行は留出組成から留出温度を露点計算 (例題 1.2) で、21~24 行は缶出組成から缶出温度を沸点計算 (例題 1.1 または例題 2.1) で算出している。最小理論段、最小還流比、Gilliland の相関については式の通り記述されているので説明を省く。

## 6. 調 湿

空気中の湿度を一定値に保つため水蒸気で加湿することは良く行なわれる操作である。

### <例題 3.7: 空気の調湿>



図の加湿器で温度  $t_a$ , [°C], 相対湿度  $\phi_a$  の空気に  $t_s = 100^\circ\text{C}$  のスチームを加え湿度  $\phi_w$  の加湿空気を得たい。所要空気、水蒸気流量を求めよ。なお、絶対湿度  $H$  [kg-H<sub>2</sub>O/kg-乾き空気] を

$$H = (18/29)p/(760 - p)$$

として、相対湿度  $\phi$  は

$$\phi = (p/p_s) \times 100$$

で与えられる。ここで、 $p$  は水蒸気分圧、 $p_s$

表 3-6 例題 3.6 の解

```

1: /* 簡易蒸留計算 */
2:
3: VAR zf(2),xf(2),yf(2),xd(2),xc(2),xw(2),yw(2)
4: VAR Pf(2),Pd(2),Pw(2),alpf(2)
5: VAR a(2),b(2),c(2)
6:
7: /* 原料中の液相の割合(フラッシュ計算) */
8: log10(Pf) = a - b / (c+tf)
9: zf = (1-q) * yf + q * xf
10: yf = (Pf/PI) * xf
11: FLASH: SUM(xf) = 1
12: RESET q # 0.5 [0,1] BY FLASH
13:
14: /* 留出温度を求める(露点計算) */
15: log10(Pd) = a - b / (c+td)
16: xd = (Pd/PI) * xc
17: DEW: SUM(xc) = 1
18: RESET td # 100 [50,150] BY DEW
19:
20: /* 缶出温度を求める(沸点計算) */
21: log10(Pw) = a - b / (c+tw)
22: yw = (Pw/PI) * xw
23: BUBBLE: SUM(yw) = 1
24: RESET tw # 100 [50,150] BY BUBBLE
25:
26: /* 最小理論段の計算 (Fenskeの式) */
27: alpd = Pd(1) / Pd(2)
28: alpw = Pw(1) / Pw(2)
29: alp = sqrt(alpd*alpw)
30: Nm + 1 = log10((xd(1)/xd(2)) *
31: (xw(2)/xw(1))) / log10(alp)
32:
33: /* 最小還流比の計算 (Underwoodの式) */
34: alpf = Pf(1) / Pf
35: UDWOOD: SUM(zf / (k-alpf)) = (1-q) / k
36: Rm = k * SUM(xd / (k-alpf)) - 1
37: RESET k # 1.1 [1,10] BY UDWOOD
38:
39: /* Gillilandの相関 (平田の式) */
40: log10((S-Sm)/(S+1))
41: = -0.9*((R-Rm)/(R+1)) - 0.17
42: Sm = Nm + 1; S = N + 1
43: R/Rm = beta
44:
45: OUTPUT Rm,Nm,R,N,q,k
46: OUTPUT td,tw,tf,xf,yf
47: OUTPUT alpf,alpd,alpw
48:
49: /* 簡易蒸留計算用データ */
50: /* n-ヘキサン-トルエン系 */
51: zf = (0.35, 0.65)
52: xd = (0.95, 0.05)
53: xw = (0.10, 0.90)
54: a = (6.91058, 6.95087)
55: b = (1189.64, 1342.31)
56: c = (226.280, 219.187)
57: tf = 94.1; PI = 760
58: beta = 1.2

```

### [ 計算結果 ]

```

Rm = 1.394307
Nm = 3.096268
R = 1.673168
N = 9.191846
q = 0.705501
k = 1.637231
td = 73.142013
tw = 103.619336
tf = 94.1
xf =
1) 0.265972      2) 0.7343
yf =
1) 0.551298      2) 0.448052
alpf =
1) 1              2) 3.397004
alpd = 3.787624
alpw = 3.250065

```

はその温度における飽和水蒸気圧 (いずれも [mmHg]) である。また、水蒸気比熱  $C_s=0.45$  [kcal/kg $\cdot$ °C] として、湿り比熱  $C_H=0.24+0.45 H$  [kcal/°C $\cdot$ kg-乾き空気] である<sup>10)</sup>。

解)  $p_s$  は温度  $T$  [°K] の関数として菅原の式<sup>11)</sup> を [mmHg] 単位に変換した式

$$\begin{aligned} & \{7.21379+(1.1520 \times 10^{-5}-4.787 \\ & \times 10^{-9} T)(T-483.16)^2\} \\ & \times \left( \frac{647.31}{T} - 1 \right) = \ln \left( \frac{1.6598 \times 10^5}{p_s} \right) \end{aligned} \quad (3-26)$$

でよく表わされる。いま、入口空気、加湿空気の体積流量を  $Air, Q$  [m<sup>3</sup>/h], スチーム流量を  $Steam$  [kg/h] とし、入口空気中の乾き空気、水蒸気量を  $Dry, stm$  [kg/h] とすれば、 $R$  を気体定数として

$$(Dry/29 + stm/18)R \cdot (t_a + 273.16) = Air \quad (3-27)$$

$$\begin{aligned} & \{Dry/29 + (Steam + stm)/18\} \cdot R(t_o + 273.16) \\ & = Q \end{aligned} \quad (3-28)$$

一方、物質収支から絶対湿度は

$$H_a = stm/Dry \quad (3-29)$$

$$H_o = (stm + Steam)/Dry \quad (3-30)$$

また、熱収支は入口空気の湿り比熱  $C_{Ha}$  を使

$$Dry \cdot C_{Ha}(t_o - t_a) + stm \cdot C_s(t_o - t_s) = 0 \quad (3-31)$$

で与えられる。これらの式を記述したリストと30°C, 40%の空気にスチームを加え90%の湿り空気を10m<sup>3</sup>/h得るときの計算例を表3-7に示す。湿度の計算はよく使用されるのでマクロとして定義してある。

## 7. 多重効用缶

蔗糖や NaCl, NaOH などの水溶液を濃縮するのに蒸発操作が行われるが、熱経済の観点から発生蒸気を再利用する多重効用缶が使用される。多重効用缶の計算は多重度に応じて問題の規模が大きくなり解き難くなるので、いくつかの計算法が提案されている<sup>12), 13)</sup>。

表3-7 例題3.7の解

```

1: /* 調 湿 */
2:
3: /*
4:   ta, phi_a -----
5:   Air  ----->|          |-----> Q
6:   Steam ----->|          |-----> to, phi_o
7:
8:   -----
9:
10:  Air   空気流量 [m3/h]
11:  Steam スチーム流量 [kg/h]
12:  Q     出口流量 [m3/h]
13:  Dry   ドライ空気 [kg/h]
14:  stm   空気中水分 [kg/h]
15:  t     温度 [°C]
16:  p     水蒸気圧 [mmHg]
17:  phi   相対湿度 [%]
18:  H     絶対湿度 [kg-H2O/kg-dry air]
19:
20: macro HUM /* 菅原の式 */
21: (7.21379+(1.152E-5 -4.787E-9*T)*(T-483.16)^2) ..
22: *(647.31/T -1) = loge(1.6598E5/p_s)
23: t + 273.16 = T
24: phi = (p / p_s)*100
25: H = (18/29)*p/(760-p)
26: end HUM
27:
28: ts = 100 ; R = 0.082 /* 気体定数 */
29: /* 入口空気の湿度 */
30: call HUM(t=ta, T=Ta, p=pa, ps=ps_a, phi=phi_a, H=Ha)
31: (Dry/29 + stm/18)*R*Ta = Air
32: Ha = stm/Dry
33:
34: /* 出口空気の湿度 */
35: call HUM(t=to, T=To, p=po, ps=ps_o, phi=phi_o, H=Ho)
36: (Dry/29 + (Steam+stm)/18)*R*To = Q
37: Ho = (Steam + stm)/Dry
38:
39: /* 熱収支 */
40: HB: Dry*CHa*(to-ta) + Steam*Cs*(to-ts) = 0
41: CHa = 0.24 + 0.45*Ha ; Cs = 0.45
42: reset to #50 [0.100] by HB
43:
44: input Q, ta, phi_a, phi_o
45: output ta, to, pa, ps_a, po, ps_o
46: output phi_a, Ha, phi_o, Ho
47: output Q, Air, Dry, stm, Steam


```

```

[ 計算結果 ]
ta = 30
to = 32.120437
pa = 12.728382
ps_a = 31.820954
po = 32.314304
ps_o = 35.904783
phi_a = 40
Ha = 0.010572
phi_o = 90
Ho = 0.027563
Q = 10
Air = 9.670263
Dry = 11.092142
stm = 0.117269
Steam = 0.188463

```

<例題3.8: 三重効用缶の設計<sup>14)</sup>>



図のような順流三重効用缶で、 $t_f=50^\circ\text{C}$ ,  $x_f=4\%$ の蔗糖水溶液( $F=5000$  kg/h)を $x_3=10\%$ まで濃縮したい。 $t_o=120^\circ\text{C}$ の加熱水蒸気にて第1缶を加熱し、第3缶の温度 $t_3$ が $60^\circ\text{C}$ に

なるよう真空度を保つ。各缶の総括伝熱係数はそれぞれ  $U_1=2000$ ,  $U_2=1400$ ,  $U_3=800$  kcal/m<sup>2</sup>/h/°C とし、各缶の伝熱面積は等しいとしてその面積  $A$  [m<sup>2</sup>] を求めよ。ただし、水溶液の比熱  $C_p$  は 1 kcal/kg/°C とし、水の蒸発潜熱  $L$  [kcal/kg] は次式で与えられる。

$$L = 538.8 + 0.625(100 - t) - 0.0005(100 - t)^2$$

また、各缶で沸点上昇は無視できるとする。

解) 全蒸気発生量を  $D$  [kg/h] とすると

$$V_1 + V_2 + V_3 = D \quad (3-32)$$

$$F(1 - x_f/x_3) = D \quad (3-33)$$

各缶における熱収支から

$$V_0 L_0 = V_1 L_1 + F C_p (t_1 - t_f) \quad (3-34)$$

$$V_1 L_1 = V_2 L_2 + (F - V_1)(t_2 - t_1) \quad (3-35)$$

$$V_2 L_2 = V_3 L_3 + (F - V_1 - V_2)(t_3 - t_2) \quad (3-36)$$

一方、伝熱速度は

$$V_0 L_0 = U_1 A (t_0 - t_1) \quad (3-37)$$

$$V_1 L_1 = U_2 A (t_1 - t_2) \quad (3-38)$$

$$V_2 L_2 = U_3 A (t_2 - t_3) \quad (3-39)$$

で表わされる。なお、 $L_i$  は温度  $t_i$  における蒸発潜熱である。この問題では  $t_1$  と  $t_2$  が知れば蒸発潜熱が計算でき、残りの変数は逐次解けるので、 $t_1$  と  $t_2$  をくり返し変数に選ぶ。

リストの計算結果を表 3-8 に示すが、蒸発潜熱の計算式をマクロとして登録使用している。このように、EQUATRAN-M によれば (3-32) から (3-39) 式を書き並べるだけで、ごく自然に解を得ることができる。

ここにあげた例題の一つづつは大した規模ではないが、それでもこれを手計算なり、プログラムを作って計算しようとするのと相当な手間がかかる。EQUATRAN-M では方程式をそのまま書 (以下、56ページへづく)

表 3-9 文献15) 例題一覧

1. 理想溶液の気液平衡関係
2. Antoine 定数の近似
3. Wilson パラメータの算出
4. NRTL 式による 2 成分液々平衡
5. 3 成分系溶解度曲線の近似
6. 熱交換器
7. McCabe-Thiele 法による蒸留計算
8. 単蒸留
9. フローシートシミュレータ

表 3-8 例題 3.8 の解

```

1: /* 多重効用缶の解析 */
2:
3: /* A 伝熱面積 [m2]
4: F 供給液量 [kg/h]
5: L 蒸発潜熱 [kcal/kg]
6: t 温度 [°C]
7: U 伝熱係数 [kcal/m2/h/°C]
8: V 蒸気流量 [kg/h]
9: */
10: macro LAMBDA /* 蒸発潜熱 */
11: L = 538.8 + 0.625*(100-t) - 0.0005*(100-t)^2
12: end LAMBDA
13:
14: /* 蒸気発生量 */
15: V1 + V2 + V3 = D
16: F*(1-xf/x3) = D
17:
18: /* 熱収支 */
19: V0*L0 = V1*L1 + F*Cp*(t1-tf)
20: V1*L1 = V2*L2 + (F-V1)*(t2-t1)
21: V2*L2 = V3*L3 + (F-V1-V2)*(t3-t2)
22:
23: /* 伝熱速度 */
24: HT1:V0*L0 = U1*A*(t0-t1)
25: HT2:V1*L1 = U2*A*(t1-t2)
26: HT3:V2*L2 = U3*A*(t2-t3)
27:
28: /* 蒸発潜熱 */
29: call LAMBDA ( L=L0 , t=t0 )
30: call LAMBDA ( L=L1 , t=t1 )
31: call LAMBDA ( L=L2 , t=t2 )
32: call LAMBDA ( L=L3 , t=t3 )
33:
34: Cp = 1
35: xf = 0.04 ; x3 = 0.1
36: U1 = 2000 ; U2 = 1400 ; U3 = 800
37: t0 = 120 ; t3 = 60
38:
39: reset t1#100 [60,120] by HT1
40: reset t2#80 [60,120] by HT3
41:
42: input F, tf
43: output F, V0, V1, V2, V3, D
44: output tf, t1, t2, t3, A

[ 入力値(=) または 初期値(#) ]
F = 5000
tf = 50
t1 = # 100
t2 = # 80

[ 計算結果 ]
F = 5000
V0 = 1421.689444
V1 = 892.398843
V2 = 988.510172
V3 = 1119.090986
D = 3000
tf = 50
t1 = 103.856819
t2 = 89.099162
t3 = 60
A = 23.167959

```

表 3-10 文献7) 例題一覧

1. Frost-Kalkwarf 式による蒸気圧の計算
2. Frost-Kalkwarf 式の係数の近似
3. van Laar 定数の推定
4. 最小二乗法による Wilson パラメータの推定
5. BWR 状態方程式による計算
6. NRTL 式による 2 成分系液々平衡
7. 理想系の沸点計算
8. 3 成分系のフラッシュ計算
9. Fenske の式
10. Underwood の式
11. Gilliland の相関
12. 簡易蒸留計算
13. Thiele-Geddes 法 (蒸留計算)



き，専門家の固有技術を活かして容易に解くことができるので，業務の効率化に役立とう。

なお，この他に文献<sup>7),15)</sup>に多数の使用例を掲げている（タイトルのみ表3—9，表3—10に示す）ので参照願いたい。

#### 参考文献

- 1) T. Westerlund, B. Scriver & S. Karrila : *C. E. S.*, Vol. 40, No. 6, p. 973 (1985).
- 2) O. A. Hougen, K. M. Watson, R. A. Ragatz : *Chemical Process Principles Part II*, McGraw-Hill, p. 1053 (1959).
- 3) R. C. Reid, J. M. Prausnitz, T. K. Sherwood : *The Properties of Gases and Liquids 3rd Ed.*, McGraw-Hill, p. 37 (1977).
- 4) W. C. Edmister : *Hydrocarbon Processing*, Vol. 47, No. 9, p. 239 (1968).
- 5) Perry's Chemical Engineers' Handbook, 4th Ed., p. 2-10 (1963).
- 6) K. Miyahara, H. Sadotomo, K. Kitamura : *J. Chem. Eng. Japan*, Vol. 3, No. 2, p. 157 (1970).
- 7) 宮原，林田，須藤：分離技術懇話会夏季研究討論会資料 (S60. 8.23).
- 8) 化学工学協会：化学工学便覧改訂四版，p. 597 (1978).
- 9) Lockhart, McHenry : *Pet. Refiner*, Vol. 37, No. 3, p. 209 (1958).
- 10) 化学工学協会：化学工学便覧改訂四版，p. 655 (1978).
- 11) 日本機械学会：蒸気表および線図，p. 3 (1955).
- 12) 藤田：化学工学 I (岩波全書)，p. 111 (1956).
- 13) 吉福：化学工学論文集，3，p. 531 (1977).
- 14) 山田：ケミカルエンジニアリング，1969年11月号，p. 95.
- 15) 宮原，林田，須藤：分離技術，Vol. 15, No. 5 (1985).
- 16) 平田，頼実：蓄留工学ハンドブック，pp 118-137, 朝倉 (1966).
- 17) 文献3)，p. 73～